类 我的文档窗口 : 窗口

	@布局配置([[根布局=真,宽度=-1,高度=-1,左内边距DP=18,右内边距DP=18,下内边距DP=40,背景颜色=-1576716]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,宽度=-1,高度DP=45,ID=1]])
	变量 相对布局2 : 相对布局
	@布局配置([[父布局=相对布局2,对齐方式=17,宽度=-1,高度=-1,内容="我的文档",粗体=真,字体大小=17,字体颜色=-16777216]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=相对布局1,宽度=-1,高度=-1,纵向布局=真,@位于某组件之下=1]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度=-1,高度=-1,显示滑块条=假,始终显示快速滚动条=假,列数=1]])
	变量 自定义宫格列表框1 : 自定义宫格列表框
	@布局配置([[父布局=相对布局1,宽度DP=50,高度DP=50,@位于布局右边=真,@位于布局底部=真,@右外边距DP=20]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,宽度=-1,高度=-1,图片资源="../../../素材/Png/93f15d7d4a6.png"]])
	变量 圆形图片框1 : 圆形图片框

	变量 文档集合:文本集合
	变量 文档数组:JSON数组=全局配置.个人数据.取JSON数组("我的文档")
	事件 我的文档窗口:创建完毕()
		订阅事件()
		本对象.屏幕方向 = 1
		本对象.显示标题栏 =假
		本对象.状态栏字体黑色 = 真
		本对象.状态栏颜色 = -1576716
		本对象.导航栏颜色 = -1576716
		加载文档项目()
	结束 事件
	方法 加载文档项目()
		循环(文档项目索引, 0, 文档数组.长度)
			文档集合.添加成员(文档数组.取文本(文档项目索引))
		结束 循环
		自定义宫格列表框1.更新项目(文档集合.长度)
	结束 方法
	事件 自定义宫格列表框1:加载布局(索引 : 整数,项目布局 : 组件容器) : 组件容器
		变量 文档项目列表布局1:文档项目列表布局
		文档项目列表布局1.文本框1.内容 = 文档集合[索引]
		/*
		假如 (文档集合[索引].分割文本("."))[1]
			是 "txt"
				文档项目列表布局1.图片框1.位图对象 = 附加资源.取位图对象(本对象,"png/文本文件.png")
		结束 假如
		*/
		返回 文档项目列表布局1
	结束 事件
	事件 自定义宫格列表框1:项目被单击(被单击组件 : 可视化组件,索引 : 整数)
		变量 操作选择弹窗:DX基础对话框
		操作选择弹窗.样式(DX样式.IOS)
		操作选择弹窗.标题("操作")
		操作选择弹窗.按钮纵向显示(真)
		操作选择弹窗.按钮1("修改内容")
		操作选择弹窗.按钮2("复制链接")
		操作选择弹窗.按钮3("删除文档")
		操作选择弹窗.显示()
		事件 操作选择弹窗:按钮2被单击() : 逻辑型
			//置剪切板文本(本对象,)
			变量 通知 : DX通知
			通知.标题("复制成功")
			通知.内容("")
			通知.成功图标()
			通知.显示()
			返回 假
		结束 事件
		事件 操作选择弹窗:按钮3被单击() : 逻辑型
			变量 删除文档网络工具1:网络工具
			删除文档网络工具1.取网页源码(全局配置.后台域名+"filedelete.php?user="+全局配置.用户账号+"&pass="+全局配置.用户密码+"&file="+文档集合[索引])
			事件 删除文档网络工具1:取网页源码结束(结果 : 文本,内容 : 字节[],cookie : 文本)
				文档数组.移除(索引)
				全局配置.个人数据.置入("我的文档",文档数组)
				全局配置.个人数据.写出到文件(取数据目录路径()+"/配置文件/个人数据/个人数据.json")
				文档集合.清空()
				自定义宫格列表框1.更新项目(文档集合.长度)
				加载文档项目()
			结束 事件
			返回 假
		结束 事件
	结束 事件

	事件 圆形图片框1:被单击()
		变量 新建文档弹窗:DX基础对话框
		变量 新建文档弹窗布局1:新建文档弹窗布局
		新建文档弹窗.自定义布局(新建文档弹窗布局1)
		新建文档弹窗布局1.选中文件名.内容=时间操作.取格式时间("yyyyMMddHHmmss")
		新建文档弹窗.样式(DX样式.IOS)
		新建文档弹窗.标题("[新建文档]")
		新建文档弹窗.按钮1("确定")
		新建文档弹窗.显示()
		事件 新建文档弹窗布局1.选中文件格式:被单击()
			变量 当前格式集合:文本集合={".txt",".html",".xls",".doc",".url"}
			变量 格式选择弹窗:DX菜单
			格式选择弹窗.依附组件(新建文档弹窗布局1.卡片布局2)
			格式选择弹窗.依附位置(DX对齐方式.下)
			格式选择弹窗.菜单列表集合(当前格式集合)
			格式选择弹窗.显示()
			事件 格式选择弹窗:列表被点击(索引 : 整数,内容 : 文本) : 逻辑型
				新建文档弹窗布局1.选中文件格式.内容=内容
				返回 假
			结束 事件
		结束 事件
		事件 新建文档弹窗:按钮1被单击() : 逻辑型
			弹出提示("等待开放")
			//创建一个文档(新建文档弹窗布局1.选中文件名.内容,新建文档弹窗布局1.选中文件格式.内容)
			返回 假
		结束 事件
	结束 事件

	变量 创建文档网络工具1:网络工具
	方法 创建一个文档(文档名:文本,文档格式:文本)
		显示加载("正在创建")
		创建文档网络工具1.取网页源码(全局配置.后台域名+"fileadd.php?user="+全局配置.用户账号+"&file="+文档名+"&filetype="+文档格式+"&pass="+全局配置.用户密码)
		事件 创建文档网络工具1:取网页源码结束(结果 : 文本,内容 : 字节[],cookie : 文本)
			编辑一个文档(文档名+文档格式)
		结束 事件
	结束 方法

	变量 编辑文档网络工具1:网络工具
	变量 新文档内容:文本
	方法 编辑一个文档(文档名:文本)
		新文档内容="欢迎使用蓝莓文档管理"
		编辑文档网络工具1.取网页源码(全局配置.后台域名+"file.php?user="+全局配置.用户账号+"&pass="+全局配置.用户密码+"&file="+文档名+"&content="+新文档内容)
		事件 编辑文档网络工具1:取网页源码结束(结果 : 文本,内容 : 字节[],cookie : 文本)
			关闭加载()
			文档数组.置入(文档名)
			全局配置.个人数据.置入("我的文档",文档数组)
			全局配置.个人数据.写出到文件(取数据目录路径()+"/配置文件/个人数据/个人数据.json")
			全局配置.个人数据=读入文本文件(取数据目录路径()+"/配置文件/个人数据/个人数据.json")
			文档数组=全局配置.个人数据.取JSON数组("我的文档")
			文档集合.清空()
			自定义宫格列表框1.更新项目(文档集合.长度)
			加载文档项目()
		结束 事件
	结束 方法


结束 类

类 新建文档弹窗布局:组件容器

	@布局配置([[根布局=真,宽度=-1,高度=-1,对齐方式=17,纵向布局=真]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,对齐方式=17,内容="文档名称",@下外边距DP=5,@上外边距DP=10]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=线性布局1,对齐方式=17,@下外边距DP=5]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,宽度DP=200,高度DP=45,圆角=45,卡片阴影=0]])
	变量 卡片布局1 : 卡片布局
	@布局配置([[父布局=卡片布局1,宽度=-1,高度=-1,背景颜色=-1973791,左内边距DP=10,右内边距DP=10,下内边距DP=3,上内边距DP=3]])
	变量 线性布局4 : 线性布局
	@布局配置([[父布局=线性布局4,宽度=-1,高度=-1,内边距=0,背景颜色=16777215,单行输入=真,对齐方式=17,字体颜色=-16777216,单行显示=真,内容="我的文档"]])
	变量 选中文件名 : 编辑框
	@布局配置([[父布局=线性布局1,对齐方式=17,内容="文档格式",@下外边距DP=5]])
	变量 文本框2 : 文本框
	@布局配置([[父布局=线性布局1,对齐方式=17,@下外边距DP=10]])
	变量 线性布局3 : 线性布局
	@布局配置([[父布局=线性布局3,宽度DP=200,高度DP=45,圆角=45,卡片阴影=0]])
	变量 卡片布局2 : 卡片布局
	@布局配置([[父布局=卡片布局2,宽度=-1,高度=-1,背景颜色=-1973791]])
	变量 线性布局5 : 线性布局
	@布局配置([[父布局=线性布局5,宽度=-1,高度=-1,对齐方式=17,内容=".txt",字体颜色=-16777216]])
	变量 选中文件格式 : 文本框

结束 类

类 文档项目列表布局:组件容器

	@布局配置([[根布局=真,宽度=-1,左内边距DP=10,右内边距DP=10,高度DP=55,背景颜色=-1576716]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,宽度=-1,高度DP=55,对齐方式=16]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度DP=30,高度DP=30,图片资源="../../../素材/附加/png/文本文件.png",@右外边距DP=8]])
	变量 图片框1 : 图片框
	@布局配置([[父布局=线性布局1,内容="文件名.txt",粗体=真,字体大小=17,字体颜色=-16777216,单行显示=真]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=相对布局1,宽度DP=10,高度DP=10,图片资源="../../../素材/Png/512222.png",@位于布局中间=真,@位于布局右边=真]])
	变量 图片框2 : 图片框

结束 类